

Optical Tamm modes

Merlyn Juárez Gutiérrez and W. Luis Mochán

September 21, 2021

1 Introduction

Guillermo Ortiz called our attention recently to Tamm modes, surface modes that exist at the interface between an homogeneous medium or vacuum and a crystal, even a 1D crystal with periodicity along the direction normal to the interface. We do a first exercise obtaining the dispersion relation of these modes.

2 Theory

2.1 System

We consider a system such as that in Fig. 1, made of a semiinfinite superlattice consisting of alternating layers a , b of dielectrics with corresponding dielectric response ϵ_a , ϵ_b and width d_a , d_b . The system is illuminated from below by a wave with wavevector $\mathbf{k}_0 = (\mathbf{Q}, k_0)$, where \mathbf{Q} lies on the interface and

$$k_0^2 = \epsilon_0 q^2 - Q^2, \quad (1)$$

with ϵ_0 the response of the ambient medium and $q = \omega/c$ the free space wavenumber. There is also a reflected wave with wavevector $\mathbf{k}_r = (\mathbf{Q}, -k_0)$. Within each medium there are upgoing and downgoing waves with wavevectors $\mathbf{k}_{\alpha\pm} = (\mathbf{Q}, \pm k_\alpha)$ with

$$k_\alpha^2 = \epsilon_\alpha q^2 - Q^2, \quad (\alpha = a, b) \quad (2)$$

2.2 Reflection amplitude

The reflection amplitude of the system may be obtained from a surface impedance formalism. Without loss of generality, we assume the interfaces




Figure 1: System consisting of a periodic superlattice made of two alternating dielectrics with response ϵ_a and ϵ_b with corresponding widths d_a and d_b . Within each media there are upwards and downwards propagating waves with wavevectors $\mathbf{k}_\pm = (Q, k_\pm)$.

are parallel to the xy plane and that $\mathbf{Q} = (0, Q, 0)$ points along the y direction. We separate the cases of TE (s) and TM (p) polarization. For s -polarization, the electric field would point along x for all the waves. Thus, we define the reflection amplitude r_s through

$$r^s = \frac{E_x^r}{E_x^i}, \quad (3)$$

where \mathbf{E}^i and \mathbf{E}^r denote the amplitudes of the incident and reflected fields. Similarly, for p -polarization the magnetic field \mathbf{H} would point along x , suggesting we define the reflection amplitude r_p through

$$r^p = \frac{H_x^r}{H_x^i}, \quad (4)$$

We define the *surface impedance* of a medium as the quotient

$$Z = \frac{E_{\parallel}}{H_{\parallel}}, \quad (5)$$

where \mathbf{E}_{\parallel} and \mathbf{H}_{\parallel} are the projections of the amplitude of the wave onto the surface. Thus, for s polarization we chose $E_{\parallel} = E_x$ and $H_{\parallel} = H_y$, while for p polarization we chose $E_{\parallel} = -E_y$ and $H_{\parallel} = H_x$. Then, using Faraday's law and Ampere-Maxwell's law we obtain the impedance for each medium

$$Z_{\alpha}^s = \frac{q}{k_{\alpha}}, \quad Z_{\alpha}^p = \frac{k_{\alpha}}{q\epsilon_{\alpha}}. \quad (\alpha = 0, a, b) \quad (6)$$

Notice that for downwards moving waves these impedances would change sign. Thus, the total fields at the surface are related through

$$\begin{aligned} E_x^s &= (1 + r^s)E_x^i, \\ H_y^s &= (1 - r^s)E_x^i/Z_0^s, \\ E_y^p &= -(1 - r^p)Z_0^p H_x^i, \\ H_x^p &= (1 + r^p)H_x^i, \end{aligned} \quad (7)$$

from which we obtain a relation between the surface impedances of the system ($Z^s = E_x^s/H_y^s$, $Z^p = -E_y^p/H_x^p$) and the reflection amplitude

$$\begin{aligned} Z^s &= \frac{1 + r^s}{1 - r^s} Z_0^s, \\ Z^p &= \frac{1 - r^p}{1 + r^p} Z_0^p, \end{aligned} \quad (8)$$

from which we obtain the reflection amplitudes

$$\begin{aligned} r^s &= \frac{Z^s - Z_0^s}{Z^s + Z_0^s}, \\ r^p &= \frac{Z_0^p - Z^p}{Z_0^p + Z^p}. \end{aligned} \tag{9}$$

The sign difference between the last two expressions is due to our definition of r^p in terms of the magnetic instead of the electric field amplitudes.

2.3 Normal modes

The normal modes of a system exist whenever the system has a response without being excited by an incoming external field. Correspondingly, the surface modes are given by the poles of the reflection amplitude. This means that the dispersion relation of the surface modes is given implicitly through the relations

$$\begin{aligned} Z_0^s + Z^s &= 0, \\ Z_0^p + Z^p &= 0. \end{aligned} \tag{10}$$

2.4 Transfer matrix

In order to apply Eqs. (10), we have to calculate the surface impedances Z^s and Z^p of the semiinfinite superlattice. To that end we first obtain the transfer matrix of a single layer. As the fields within one homogeneous layer is given by the sum of an upwards and a downwards moving wave, each characterized by a perpendicular wavenumber ($\pm k_\alpha$), field ($E_{\alpha\pm}^s(z)$ for s and $H_{\alpha\pm}^p(z)$ for p polarization) and surface impedance $\pm Z_\alpha$. The parallel components of the electric and magnetic fields $E_{\alpha\parallel}(z)$ and $H_{\alpha\parallel}(z)$ are related to the upgoing and downgoing fields through

$$\begin{aligned} E_{\alpha\parallel}^s(z) &= E_{\alpha+}^s(z) + E_{\alpha-}^s(z), \\ H_{\alpha\parallel}^s(z) &= (E_{\alpha+}^s(z) - E_{\alpha-}^s(z))Y_\alpha^s, \\ E_{\alpha\parallel}^p(z) &= (H_{\alpha+}^p(z) - H_{\alpha-}^p(z))Z_\alpha^p, \\ H_{\alpha\parallel}^p(z) &= (H_{\alpha+}^p(z) + H_{\alpha-}^p(z)), \end{aligned} \tag{11}$$

which we write as the matrix equations

$$\begin{pmatrix} E_{\alpha\parallel}^s \\ H_{\alpha\parallel}^s \end{pmatrix}_z = \begin{pmatrix} 1 & 1 \\ Y_\alpha^s & -Y_\alpha^s \end{pmatrix} \begin{pmatrix} E_{\alpha+}^s \\ E_{\alpha-}^s \end{pmatrix}_z, \quad (12)$$

$$\begin{pmatrix} E_{\alpha\parallel}^p \\ H_{\alpha\parallel}^p \end{pmatrix}_z = \begin{pmatrix} Z_\alpha^p & -Z_\alpha^p \\ 1 & 1 \end{pmatrix} \begin{pmatrix} E_{\alpha+}^p \\ E_{\alpha-}^p \end{pmatrix}_z,$$

where we introduced the surface admittance $Y_\alpha^\zeta \equiv 1/Z_\alpha^\zeta$ for each polarization $\zeta = s, p$. Together with $E_\pm^\zeta(z) = E_\pm^\zeta(z')e^{\pm ik_\alpha(z-z')}$ for any two positions within the layer, allows us to write the fields at the upper interface of a layer (u) in terms of the fields at its lower interface (l),

$$\begin{pmatrix} E_{\alpha\parallel}^\zeta \\ H_{\alpha\parallel}^\zeta \end{pmatrix}_u = M_\alpha^\zeta \begin{pmatrix} E_{\alpha\parallel}^\zeta \\ H_{\alpha\parallel}^\zeta \end{pmatrix}_l \quad (13)$$

where

$$\begin{aligned} M_\alpha^s &= \begin{pmatrix} 1 & 1 \\ Y_\alpha^s & -Y_\alpha^s \end{pmatrix} \begin{pmatrix} e^{ik_\alpha d_\alpha} & 0 \\ 0 & e^{-ik_\alpha d_\alpha} \end{pmatrix} \begin{pmatrix} 1 & 1 \\ Y_\alpha^s & -Y_\alpha^s \end{pmatrix}^{-1}, \\ M_\alpha^p &= \begin{pmatrix} Z_\alpha^p & -Z_\alpha^p \\ 1 & 1 \end{pmatrix} \begin{pmatrix} e^{ik_\alpha d_\alpha} & 0 \\ 0 & e^{-ik_\alpha d_\alpha} \end{pmatrix} \begin{pmatrix} Z_\alpha^p & -Z_\alpha^p \\ 1 & 1 \end{pmatrix}^{-1}, \end{aligned} \quad (14)$$

yielding

$$M_\alpha^\zeta = \begin{pmatrix} \cos k_\alpha d_\alpha & iZ_\alpha^\zeta \sin k_\alpha d_\alpha \\ iY_\alpha^\zeta \sin k_\alpha d_\alpha & \cos k_\alpha d_\alpha \end{pmatrix}, \quad (15)$$

which has the same form for both polarizations. From the transfer matrix of each layer, we may obtain the transfer matrix of a period as a simple matrix multiplication,

$$M^\zeta = M_b^\zeta M_a^\zeta, \quad (16)$$

as the parallel fields E_\parallel and H_\parallel are continuous across each interface. Notice that by construction, the determinant of the transfer matrix is 1.

2.5 Bulk modes

Within a periodic system, the normal modes may be written as a superposition of Bloch modes, which acquire a phase e^{ipd} when translated across a period of width $d = d_a + d_b$, where p is Bloch's vector. Thus

$$\begin{pmatrix} E_\parallel^\zeta \\ H_\parallel^\zeta \end{pmatrix}_d = M^\zeta \begin{pmatrix} E_\parallel^\zeta \\ H_\parallel^\zeta \end{pmatrix}_0 = e^{ip^\zeta d} \begin{pmatrix} E_\parallel^\zeta \\ H_\parallel^\zeta \end{pmatrix}_0, \quad (17)$$

i.e., $e^{ip^\zeta d}$ is an *eigenvalue* of the transfer matrix. Notice that for an infinite system, p^ζ ought to be real in order to avoid divergences at $z = \pm\infty$. However, for semiinfinite systems we may also allow for complex Bloch's vector as long as its imaginary part is positive, so that the mode decays as $z \rightarrow \infty$. The characteristic equation

$$\det(M^\zeta - \Lambda^\zeta) = (\Lambda^\zeta)^2 - \Lambda^\zeta \text{tr} M + \det M, \quad (18)$$

for the eigenvalue $\Lambda^\zeta = e^{ip^\zeta d}$ may be simplified to

$$\cos p^\zeta d = \frac{1}{2} \text{tr} M^\zeta, \quad (19)$$

as $\det M^\zeta = 1$.

As \cos is an even function, for any solution p there is also a solution $-p$. To identify the correct choice of a solution we may add a small imaginary part to the dielectric functions, to ensure that p is complex, and choose that value that has a positive imaginary part.

2.6 Surface impedance

The surface impedance of the system may now be obtained from the eigenvector of M^ζ ,

$$(M^\zeta - \Lambda^\zeta) \begin{pmatrix} E_{\parallel}^\zeta \\ H_{\parallel}^\zeta \end{pmatrix}_0, \quad (20)$$

from which we can read the surface impedance

$$\begin{aligned} Z^\zeta &= \frac{M_{12}^\zeta}{\Lambda^\zeta - M_{11}^\zeta} = \frac{\Lambda^\zeta - M_{22}^\zeta}{M_{21}^\zeta} \\ &= 2 \frac{M_{12}^\zeta}{M_{22}^\zeta - M_{11}^\zeta \pm \sqrt{4 - (M_{11}^\zeta + M_{22}^\zeta)^2}} \\ &= \frac{1}{2} \frac{M_{11}^\zeta - M_{22}^\zeta \pm \sqrt{4 - (M_{11}^\zeta + M_{22}^\zeta)^2}}{M_{21}^\zeta}. \end{aligned} \quad (21)$$

The signs of the square roots should be chosen in such a way that the surface impedance has a positive real part, so that energy is dissipated (even if only infinitasimally for transparent systems) within the superlattice.

2.7 Summary

We have now all the ingredients for the calculation of the reflection amplitude and its poles, which yield the normal modes. For example, we could give Q and ω and the polarization ζ , calculate all wavevectors from Eqs. (1) and (2), the surface impedance of each medium through Eq. (6), the transfer matrix of each layer and the complete transfer matrix through Eqs. (15) and (16), Bloch's vector through Eq. (19), the surface impedance of the superlattice from Eq. 21 and finally the reflection amplitude and/or dispersion relations from Eqs. (9) and (10).

3 Implementation

In this section we build a computational code to perform the calculations summarized above for a dispersionless dielectric system. We use the *Perl Data Language* PDL.

First we include the required pragmas and packages.

```
# Calculate and plot the reflection amplitude
# of a dielectric superlattice
use warnings;
use strict;
use v5.12;
use Getopt::Long;
use List::Util;
use PDL;
use PDL::NiceSlice;
use PDL::Graphics::Gnuplot;
```

Figure 2: packages

Now we initialize the required parameters. I will use units such that the period is $d = 1$ and the speed of light is $c = 1$.

```
my $eps0=1; # dielectric function of the ambient. Vacuum by default
my ($epsa, $epsb); # dielectric function of the layers
my ($da, $db); # widths of layers a and b
my $im=1e-7; # small imaginary part added to disambiguate
my $Qmin=0; # minimum wavevector
my $Qmax; # maximum wavevector
my $NQ; # number of different wavevectors
```

```

my $wmin=0; # minimum frequency
my $wmax; # maximum frequency
my $Nw;    # number of frequencies
my $pol;   # polarization
my $filename; # filename for plot
my $title=''; # title for plot
my $cbrange;
my $options=q(
    'eps0=f'=>\$eps0, # dielectric function of ambient
    'epsa=f'=>\$epsa, # dielectric function of layers a
    'epsb=f'=>\$epsb, # dielectric function of layers b
    'im=f'=>\$im, # imaginary part to add to epsilons
    'da=f'=>\$da, # width of layers a
    #'db=f'=>\$db, # width of layers b, d-da
    'Qmin=f'=>\$Qmin, # minimum wavevector
    'Qmax=f'=>\$Qmax, # maximum wavevector
    'NQ=i'=>\$NQ, # number of wavevectors
    'wmin=f'=>\$wmin, # minimum frequency
    'wmax=f'=>\$wmax, # maximum frequency
    'Nw=i'=>\$Nw, # number of frequencies
    'pol=s'=>\$pol, # polarization
    'filename=s'=>\$filename, # filename for plot
    'title=s'=>\$title, # title for plot
    'cbrange=s'=>\$cbrange, # optional cbrange "[min,max]"
);
my %options=(eval $options);
die "Bad option definition; $$" if $@;
GetOptions(%options) or usage($options, "Bad options");
usage($options, "Undefined parameters")
    unless List::Util::all
        {defined $_}
        ($eps0, $epsa, $epsb, $im, $da, $Qmin, $Qmax, $NQ,
         $wmin, $wmax, $Nw, $pol, $filename, $title);

```

Figure 3: initialization

We need a routine to print a usage message and quit when a mistake is found.

```

sub usage {
    my ($options, $message)=@_;

```



```

    say $message if defined $message;
    say $options;
    exit 1;
}

```

Figure 4: usage

We check the input parameters and set the remaining parameter(s).

```

usage($options, "Parameter da should obey 0<=da<=1: $da")
    unless 0<=$da<=1;
$db=1-$da; # enforce units so that d=d_a+d_b=1
my %pols=(te=>'s', s=>'s', p=>'p', tm=>'p');
my $polarization=$pols{lc $pol}; # normalize polarization
usage($options, "Polarization should be s, p, TE or TM: $pol")
    unless defined $polarization;
my ($ceps0, $cepsa, $cepsb)
    =map {$_+$im*i()} ($eps0, $epsa, $epsb); # add imaginary parts

```

Figure 5: check-parameters

Now we define the wavevector and frequency (Eqs. (1) and (2)) ndarrays and apply our formulas above. I use dummy arrays so I can thread over all wavevectors and frequencies.

```

my $Q=zeros($NQ)->xlinvals($Qmin,$Qmax)->dummy(1); # Q,w
my $w=zeros($Nw)->xlinvals($wmin,$wmax)->dummy(0); # Q,w
my $q=$w; # assume c=1
my ($k0, $ka, $kb)=map {mysqrt($_*$q**2-$Q**2)} ($ceps0, $cepsa, $cepsb);

```

Figure 6: setranges

I introduce a square root with branch cut just below the positive real axis, so it yields a non-negative imaginary part. I use `thread_define` so that it can be threaded over arrays.

```

sub mysqrt {
    my $a=shift;
    my $b=null;
    _mysqrt($a, $b);
    return $b;
}

```

```

BEGIN{
  thread_define '_mysqrt(a());[o]b())'=> over {
    my ($a, $b)=@_;
    my $r=sqrt($a);
    $r= -$r if $r->im<0;
    $b.=$r;
  };
}

```

Figure 7: mysqrt

I calculate the surface impedance (Eqs. (6)),

```

my ($Z0, $Za, $Zb)=$polarization eq 's' # Q,w
? ($q/$k0, $q/$ka, $q/$kb)
: ($k0/($ceps0*$q), $ka/($cepsa*$q), $kb/($cepsb*$q));

```

Figure 8: surface-impedance

the transfer matrices (Eqs. (15)),

```

my ($Ma, $Mb)=map { # n,m,Q,w
  my ($kd,$Z)=@$_;
  pd1([[cos($kd), i()*$Z*sin($kd)], [i()/ $Z*sin($kd), cos($kd)]])
    ->mv(-1,0)->mv(-1,0); #m,n,Q,w
} ([$ka*$da, $Za], [$kb*$db, $Zb]);

```

Figure 9: transfer-matrices

the total transfer matrix (Eq. (16)),

```

my $M=$Mb x $Ma; # n,m,Q,w

```

Figure 10: total-transfer-matrix

and the eigenvalue (Eq. (19)).

```

my $cospd=($M((0),(0))+ $M((1),(1)))/2; #Q,w
my $eipd=eipd($cospd); # Q,w

```

Figure 11: eigenvalue

To get the eigenvalue I write a function `eipd` that takes the cosine and builds the appropriate complex exponential. I use again `thread_define` so that I write the routine for a scalar but then iterate implicitly with the threading machine.

```

sub eipd {
  my $cospd=shift;
  my $eipd=null;
  _eipd($cospd,$eipd);
  return $eipd;
}

BEGIN {
  thread_define '_eipd(c();[o]r())'=> over {
    my ($c, $r)=@_;
    my $s=sqrt(1-$c**2);
    my $eipd=$c+$s*i();
    my $mod2=$eipd->abs2;
    $eipd=$c-$s*i() if $mod2>1;
    $eipd=$c-$s*i() if $mod2==1 && $s->re < 0; # ??
    $r.=$eipd;
  };
}

```

Figure 12: eipd

With the transfer matrix and its eigenvalue I can finally calculate the surface impedance and the reflection amplitude (give or take a sign, Eqs. (9))

```

# Notice that M22->$M((1),(1)) and M21->$M((0),(1))
# as the first index of a pdl designates columns, not rows
my $Z=($eipd-$M((1),(1)))/$M((0),(1)); # Q,w
my $r=($Z0-$Z)/($Z0+$Z); # Q,w

```

Figure 13: reflection

Finally, I plot the result. As I normalized distances with d and took $c = 1$, my \$variable Q actually means Qd and ω is actually $\omega d/c$.

```

$filename=".png" unless $filename=~m/\.png$/;
my $gpw=gpwin('png', output=>"$filename");
my %plotoptions=(title=>$title, xlabel=>'Wavevector Qd',
                  ylabel=>'Frequency {/symbol w} d/c');
$plotoptions{cbrange}=pdl($cbrange) if defined $cbrange;
$gpw->plot(\%plotoptions, with=>'image', $Q, $w, $r->abs);

```

Figure 14: plot

The full program is thus

```
<<packages>>
<<initialization>>
<<check-parameters>>
<<setranges>>
<<surface-impedance>>
<<transfer-matrices>>
<<total-transfer-matrix>>
<<eigenvalue>>
<<reflection>>
<<plot>>
<<usage>>
<<mysqrt>>
<<eipd>>
```

4 Results

I run the program to get the magnitude of the reflection amplitude as a function of Q and ω .

4.1 Transverse electric

```
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-2 --da .5 \
--Qmax 10 --NQ 500 --wmax 10 --Nw 500 --pol te \
--filename te \
--title "Transverse electric e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-2" \
--cbrange "[0,5]"
```

4.2 Transverse magnetic

```
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-3 --da .5 \
--Qmax 10 --NQ 500 --wmax 10 --Nw 500 --pol tm \
--filename tm \
--title "Transverse magnetic e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-3" \
--cbrange "[0,5]"
```




Figure 15: Reflection amplitude for transverse electric polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$.




Figure 16: Reflection amplitude for transverse magnetic polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$.

4.3 Discussion

As can be seen from the plots, there is a nice sequence of Tamm modes. we reduced the `cbrange` scale to make the structure visible, but I expect the field amplification may be as large as the inverse of the fictitious imaginary part I added to the ϵ 's. As the modes seem to be narrow, I do a closeup below.

```
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-3 --da .5 \  
  --Qmin 5 --Qmax 6 --NQ 500 \  
  --wmin 5 --wmax 6 --Nw 500 \  
  --pol te --filename te_cu \  
  --title "Transverse electric e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-3" \  
  --cbrange "[0,10]"
```




Figure 17: Reflection amplitude for transverse electric polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$.

```
./reflection.pl --eps0 1 --epsa 2 --epsb 3 --im 1e-3 --da .5 \  
  --Qmin 7 --Qmax 8 --NQ 500 \  
  --wmin 7 --wmax 8 --Nw 500 \  
  --pol te --filename te_cu \  
  --title "Transverse electric e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-3" \  
  --cbrange "[0,10]"
```

```

--wmin 6 --wmax 7 --Nw 500 --pol tm \
--filename tm_cu \
--title "Transverse magnetic e0=1 ea=2 eb=3 da=.5 db=.5 im=1e-3" \
--cbrange "[0,10]"

```




Figure 18: Reflection amplitude for transverse magnetic polarization, for a system with $\epsilon_0 = 1$, $\epsilon_a = 2$, $\epsilon_b = 3$, $d_a = 0.5$ and $d_b = 0.5$.

5 Summary

Through a simple transfer matrix formalism we obtained the transfer matrix and from it the surface impedance, reflection amplitude and surface modes of a dispersionless dielectric superlattice. We found Tamm modes for both TE and for TM polarizations, though the modes seem to couple more strongly in the TE than in the TM case, i.e., better tuning would be required to excite them in the TE case, though the amplification of the field is dominated by the artificial imaginary part added to the response and without dissipation it could reach infinity. The modes only exist outside of the light cone of the ambient, as could have been expected, and arise from the photonic bandgaps

of the superlattice. We still have to explore how the dispersion relations depend on tunable parameters of the system such as the relative widths of the layers and their indices of refraction.

6 Appendix

6.1 Metapost code

I include below the metapost code used to produce Fig. 1.

```

g_use_svg = 0; %0 for postscript, 1 for svg
prologues:=3;
u=1cm;

outputtemplate := "%j-%c.eps";
if g_use_svg > 0:
    outputtemplate := "%j-%c.svg";
    outputformat := "svg";
fi
beginfig(1)
    for i=0 upto 4:
        fill unitsquare xscaled 5u yscaled .5u shifted (0,i*u)
            withcolor .5green+.5white;
        fill unitsquare xscaled 5u yscaled .5u shifted (0,.5u)
            shifted (0,i*u) withcolor .5blue+.5white;
        drawarrow((0,-.125u)--(.7u,.125u)) shifted(-.7u,0)
            shifted(-.2u,0) shifted (2.5u,0) shifted (0,.25u)
            shifted(0,i*u);
        drawarrow((0,.125u)--(.7u,-.125u))
            shifted(.2u,0) shifted (2.5u,0) shifted (0,.25u)
            shifted(0,i*u);
        drawarrow((0,-.20u)--(.7u,.20u)) shifted(-.7u,0)
            shifted(-.2u,0) shifted (2.5u,0) shifted (0,.75u)
            shifted(0,i*u);
        drawarrow((0,.20u)--(.7u,-.20u))
            shifted(.2u,0) shifted (2.5u,0) shifted (0,.75u)
            shifted(0,i*u);
    endfor;
    drawarrow((0,-.1u)--(.7u,.1u)) shifted(-.7u,0)
        shifted(-.2u,0) shifted (2.5u,0) shifted (0,-.2u);

```

```

drawarrow((0,.1u)--(.7u,-.1u))
  shifted(.2u,0) shifted (2.5u,0) shifted (0,-.2u);
label(btex $\epsilon_a$ etex scaled 1.5,(1u,.25u));
label(btex $\epsilon_b$ etex scaled 1.5,(1u,.75u));
interim aangle:=180;
drawdblarrow ((0,0)--(0,.5u)) shifted(0.2u,0) shifted (5u,0);
label.rt(btex $d_a$ etex scaled 1.5,(0,.25u) shifted (0.2u,0) shifted (5u,0));
drawdblarrow ((0,0)--(0,.5u)) shifted(0.2u,0.5u) shifted (5u,0);
label.rt(btex $d_b$ etex scaled 1.5,(0,.75u) shifted (0.2u,0) shifted (5u,0));
endfig;
end.

```

I run the metapost code as shown below, and it produces a postscript file named `notas-1.eps`.

```
mpost notas
```