

Esparcimiento de Mie en 2D

Luis Mochán

9 de octubre de 2021

Consideremos un cilindro circular de radio R hecho de algún material con respuesta dieléctrica ϵ sobre el que incide una onda plana monocromática de frecuencia ω . Orientemos el eje z a lo largo del eje del cilindro y el eje x a lo largo de la dirección de propagación, como en la figura 1.

1. Caso transversal magnético (TM)

1.1. Campo incidente

Consideremos primero el caso en que el campo magnético de la onda incidente \mathbf{H}_i apunta en la dirección z . Por simetría, el campo magnético de todas las ondas esparcidas también apuntaría en la dirección z . El campo eléctrico estaría en el plano xy . El campo incidente puede escribirse como

$$\mathbf{H}_i = H_{i0} \hat{\mathbf{z}} e^{i(qx - \omega t)}, \quad (1)$$

donde $q = \omega/c$ y c es la velocidad de la luz. Normalizamos la amplitud del campo incidente, $H_{i0} = 1$, ignoramos la dependencia temporal de los campos, pues todos tienen la exponencial $e^{-i\omega t}$, la cual puede cancelarse, y expresamos $e^{iqx} = e^{iqr \cos \theta}$ en términos de funciones de Bessel y eigenfunciones del ímpetu angular \hat{L}_z

$$e^{iqx} = \sum_l i^l J_l(qr) e^{il\theta}, \quad (2)$$

donde r y θ son las coordenadas polares radial y angular, l es el eigenvalor del ímpetu angular y J_l son las funciones de Bessel del primer tipo.

1.2. Campo en el exterior

En el vacío la componente z del campo magnético obedece la ecuación de onda, que para ondas monocromáticas se convierte en la ecuación de




Figura 1: Sistema consistente en un cilindro de radio R sobre el que incide una onda plana con vector de onda \mathbf{q} propagándose en la dirección x , y produciendo ondas esparcidas.

Helmholtz,

$$(\nabla^2 + q^2)H = 0. \quad (3)$$

En coordenadas cilíndricas, el laplaciano puede escribirse como

$$\nabla^2 = \frac{1}{r} \frac{\partial}{\partial r} r \frac{\partial}{\partial r} + \frac{1}{r^2} \frac{\partial^2}{\partial \theta^2}, \quad (4)$$

y la ecuación de Helmholtz se puede resolver por separación de variables $H(r, \theta) = f(r)g(\theta)$. La solución de la parte angular con condiciones periódicas $g(\theta) = g(\theta + 2\pi)$ es de la forma $g(\theta) = e^{il\theta}$ con l entero, y la ecuación correspondiente para la parte radial es

$$\frac{1}{r} \frac{\partial}{\partial r} r \frac{\partial}{\partial r} f(r) - \frac{l^2}{r^2} f(r) + q^2 f(r) = 0, \quad (5)$$

que se puede reescribir como

$$\chi^2 f''(\chi) + \chi f'(\chi) + (\chi^2 - l^2) f(\chi) = 0, \quad (6)$$

donde hicimos el cambio de variable $\chi \equiv qr$ y donde f'' y f' denotan las derivadas de f respecto a su argumento. Identificamos esta como la ecuación de Bessel y sus soluciones son alguna combinación de ondas de Bessel correspondientes al momento angular l . Como era de esperarse, cada uno de los sumandos en la expansión (2) satisface la ec. de onda. El campo producido por el cilindro debe consistir de ondas salientes en el límite $r \rightarrow \infty$, y por tanto el campo total en el vacío debe ser de la forma

$$H(r, \theta) = \sum_l i^l (J_l(qr) + a_l H_l^{(1)}(qr)) e^{il\theta}, \quad r > R \quad (7)$$

donde $H_l^{(1)}(qr)$ son las funciones de Hankel salientes y a_l son coeficientes a determinar.

1.3. Campo en el interior

En el interior del cilindro también se cumple una ecuación de onda análoga a la ec. (3) pero sustituyendo q por el número de onda k dentro del material, donde

$$k^2 = \epsilon q^2. \quad (8)$$

La solución en el interior debe ser regular en el origen, y por lo tanto puede escribirse como

$$H(r, \theta) = \sum_l i^l b_l J_l(kr) e^{il\theta}, \quad r < R \quad (9)$$

donde b_l son coeficientes a determinar.

1.4. Condiciones de frontera

Para obtener los coeficientes a_l y b_l recurrimos a las condiciones de frontera en la superficie $r = R$. El campo magnético es tangencial a la superficie del cilindro y por tanto debe ser continuo, con el mismo valor en R^+ que en R^- . Por tanto

$$\sum_l i^l (J_l(qR) + a_l H_l^{(1)}(qR)) e^{il\theta} = \sum_l i^l b_l J_l(kR) e^{il\theta}. \quad (10)$$

Como las funciones trigonométricas $e^{il\theta}$ forman una base ortogonal del espacio de funciones angulares y son linealmente independientes, podemos eliminarlas junto con la suma en la ec. (10) y simplificar la condición de frontera escribiéndola como

$$(J_l(qR) + a_l H_l^{(1)}(qR)) = b_l J_l(kR). \quad (11)$$

Empleando la ec. de Ampère-Maxwell podemos hallar el desplazamiento eléctrico \mathbf{D} a partir del campo magnético \mathbf{H} . La componente tangencial del desplazamiento es entonces

$$D_\theta = \frac{1}{q} \frac{\partial}{\partial r} H. \quad (12)$$

Como la componente tangencial del campo eléctrico es continua en R , y $\mathbf{E} = \mathbf{D}$ en el exterior y $\mathbf{E} = \mathbf{D}/\epsilon$ en el interior del material, entonces

$$\epsilon q (J'_l(qR) + a_l (H^{(1)})'_l(qR)) = k b_l J'_l(kR). \quad (13)$$

1.5. Esparcimiento

Podemos eliminar b_l de las ecuaciones (11) y (13) para obtener finalmente los coeficientes de esparcimiento a_l

$$a_l = - \frac{\epsilon q J'_l(qR) J_l(kR) - k J_l(qR) J'_l(kR)}{\epsilon q (H^{(1)})'_l(qR) J_l(kR) - k H_l^{(1)}(qR) J'_l(kR)} \quad (14)$$

2. Caso transversal eléctrico (TE)

En este caso el campo eléctrico apunta en la dirección z y en analogía con el caso TM, escribimos

$$E(r, \theta) = \sum_l i^l (J_l(qr) + a_l H_l^{(1)}(qr)) e^{il\theta}, \quad r > R \quad (15)$$

$$E(r, \theta) = \sum_l i^l b_l J_l(kr) e^{il\theta}, \quad r < R \quad (16)$$

en analogía a las ecs. (7) y (9). El magnético paralelo puede obtenerse de la ley de Faraday, por lo que

$$B_\theta = \frac{1}{q} \frac{\partial}{\partial r} E. \quad (17)$$

Tanto E como B_θ son continuos en R , lo cual lleva a las condiciones de contorno

$$(J_l(qR) + a_l H_l^{(1)}(qR)) = b_l J_l(kR). \quad (18)$$

y

$$q(J_l'(qR) + a_l (H_l^{(1)})'_l(qR)) = k b_l J_l'(kR). \quad (19)$$

Resolviendo, podemos obtener las amplitudes de dispersión

$$a_l = - \frac{q J_l'(qR) J_l(kR) - k J_l(qR) J_l'(kR)}{q (H_l^{(1)})'_l(qR) J_l(kR) - k H_l^{(1)}(qR) J_l'(kR)} \quad (20)$$

3. Secciones transversales

Para distancias grandes, las funciones de Bessel pueden aproximarse por

$$H_l^{(1)}(qr) = \sqrt{\frac{2}{\pi qr}} e^{i(qr - (l+1/2)\pi/2)}, \quad (21)$$

Por lo tanto, el flujo de energía esparcida es

$$S = \frac{c}{4\pi^2 qr} \text{Re} \sum_{l,l'} a_l a_{l'}^* e^{i(l-l')\theta} e^{-i(l-l')\pi/2} \quad (22)$$

Multiplicando por r obtenemos la potencia esparcida por unidad de longitud (axial) y por unidad de ángulo, y dividiendo entre la intensidad incidente

$$I_i = \frac{c}{8\pi}, \quad (23)$$

obtenemos la sección diferencial de esparcimiento

$$\frac{d\sigma_{\text{es}}}{d\theta} = \frac{2}{\pi q} \text{Re} \sum_{l,l'} a_l a_{l'}^* e^{i(l-l')\theta} \quad (24)$$

Como el problema es 1D esta *sección* tiene unidades de longitud, no de área. Integrando sobre el ángulo obtenemos la sección total de esparcimiento.

$$\sigma_{\text{es}} = \frac{4}{q} \sum_l |a_l|^2. \quad (25)$$

En la dirección frontal hay interferencia entre el campo incidente y el campo esparcido. Escribimos el campo total F

$$F(\mathbf{r}) = e^{iqx} + \frac{f(\theta)e^{iqr}}{\sqrt{r}}, \quad (26)$$

donde identificamos a F con el campo magnético para el caso TM y con el campo eléctrico para el caso TE, donde

$$f(\theta) = \sqrt{\frac{2}{\pi q}} \sum_l i^l a_l e^{il\theta} e^{-i(l+1/2)\pi/2} = \sqrt{\frac{2}{i\pi q}} \sum_l a_l e^{il\theta} \quad (27)$$

Para ángulos pequeños hacemos las aproximaciones

$$f(\theta) \approx f(0) = \sqrt{\frac{2}{i\pi q}} \sum_l a_l, \quad (28)$$

$e^{iqr} \approx e^{iqx} e^{iqy^2/2x}$. Si colocamos un detector en la dirección frontal, la potencia por unidad de longitud que recibiría sería

$$P = \frac{1}{8\pi} \left(\ell + 2\text{Re} \frac{f(0)}{\sqrt{x}} \int_{-\ell/2}^{\ell/2} dy e^{iqy^2/2x} \right) + \dots \quad (29)$$

donde ℓ es la apertura del detector y el término que despreciamos es del orden de $1/\sqrt{qr}$ veces menor que el que conservamos. Suponiendo que el detector cubre un ángulo pequeño, $\ell \ll x$, pero que es tan ancho que $e^{iqy^2/2x}$ tiene muchas oscilaciones, en su interior, i.e. $\ell l^2 \gg \lambda x$, entonces

$$P = \frac{c}{8\pi} \left(\ell + 2\text{Re} f(0) \sqrt{\frac{2\pi i}{q}} \right), \quad (30)$$

donde usamos la integral

$$\int_{-\infty}^{\infty} d\zeta e^{i\zeta^2} = \sqrt{i\pi}, \quad (31)$$

como se puede verificar en el apéndice 8. El primer término en la ec. (30) es la potencia que llegaría al detector de no haber partícula. Por lo tanto, el cociente del segundo al primero puede identificarse en términos de la fracción de energía que no llega al detector debido a la presencia de la partícula, i.e., a la sección transversal de extinción,

$$\sigma_{\text{ex}} = -2\text{Re} f(0) \sqrt{\frac{2\pi i}{q}} = -\frac{4}{q} \text{Re} \sum_l a_l. \quad (32)$$

La sección transversal de absorción está dada por la diferencia

$$\sigma_{\text{abs}} = \sigma_{\text{ex}} - \sigma_{\text{es}}. \quad (33)$$

4. Campo superficial.

Para evaluar el campo dentro del cilindro debemos obtener los coeficientes b_l . Empleando las ecs. (11) o (18) obtenemos para ambas polarizaciones

$$b_l = \frac{J_l(qR) + a_l H_l^{(1)}(qR)}{J_l(kR)}. \quad (34)$$

El campo eléctrico tangencial superficial puede evaluarse a partir de las ecs. (12) para polarización TM y de (16) para polarización TE usando condiciones de continuidad. Podemos obtener una medida de su magnitud promediando su cuadrado sobre la superficie,

$$\langle E_\theta^2 \rangle = \left| \frac{k}{\epsilon} \right|^2 \sum_l |b_l J'_l(kR)|^2, \quad \text{TM} \quad (35)$$

$$\langle E^2 \rangle = \sum_l |b_l J_l(kR)|^2. \quad \text{TE} \quad (36)$$

5. Límite no retardado

En el caso no retardado $qR \ll 1$ puedo obtener resultados aproximados tomando los valores asintóticos de las funciones de Bessel,

$$\begin{aligned} J_l(\zeta) &\approx \frac{\zeta^l}{2^l l!} \\ N_0(\zeta) &\approx \frac{2}{\pi} \log \zeta \\ N_l(\zeta) &\approx -\frac{(l-1)!}{\pi} \left(\frac{2}{\zeta} \right)^l, \quad l > 0 \end{aligned} \quad (37)$$

Sustituyendo en la ec. (14) obtenemos para $l > 1$

$$a_l = i\pi(qR)^{2l} \frac{l}{(2^l l!)^2} \frac{\epsilon - 1}{\epsilon + 1}. \quad (38)$$

Multiplicando por la función de Hankel correspondiente, obtenemos el campo magnético inducido en el exterior, dado por

$$H^{\text{ind}}(r) = \sum_l (iq)^l \frac{\epsilon - 1}{\epsilon + 1} \frac{1}{2^l l!} \frac{R^{2l}}{r^l} e^{il\theta} \approx iq \frac{\epsilon - 1}{\epsilon + 1} R^2 \frac{x}{r^2}, \quad (39)$$

dominado por las contribuciones $l = \pm 1$. Aquí he ignorado el término con $l=0$ pues no contribuye al campo eléctrico. El campo eléctrico inducido puede obtenerse de la ec. de Ampère-Faraday

$$\mathbf{E}^{\text{ind}} = -\frac{1}{iq} \nabla \times \mathbf{H}^{\text{ind}} = -\frac{\epsilon - 1}{\epsilon + 1} R^2 \left(\frac{\partial}{\partial y} \frac{x}{r^2}, \frac{\partial}{\partial x} \frac{x}{r^2} \right). \quad (40)$$

Notamos que $x/r^2 = \partial \log r / \partial x$ y que $\partial^2 \log r / \partial x^2 = -\partial^2 \log r / \partial y^2$, pues $\log r$ obedece la ec. de Laplace, por lo que podemos reescribir la ec. (40) como

$$\mathbf{E}^{\text{ind}} = -2\mathbf{p} \cdot \nabla \nabla \log r = (2\mathbf{p} \cdot \mathbf{r} \mathbf{r} - \mathbf{p} r^2) / r^4, \quad (41)$$

donde identificamos el dipolo inducido por unidad de longitud

$$\mathbf{p} = \alpha \mathbf{E}^{\text{ex}}, \quad (42)$$

donde el campo eléctrico externo tiene la misma magnitud que el campo magnético externo (que hemos normalizado a 1) y apunta en la dirección y , y donde identificamos la polarizabilidad por unidad de longitud

$$\alpha = \frac{\epsilon - 1}{\epsilon + 1} \frac{R^2}{2}. \quad (43)$$

Este es un resultado bien conocido que puede obtenerse mediante un desarrollo multipolar cuasiestático.

6. Código

6.1. Dispersión

A continuación presento un código para el cálculo de la dispersión.

Empiezo el código cargando los paquetes y las opciones usuales.

```
use v5.12;
use warnings;
use List::Util;
use Getopt::Long;
use IO::Prompter;
use PDL;
use PDL::NiceSlice;
use PDL::Graphics::Gnuplot;
```

Figura 2: paquetes

Declara e inicializa los parámetros.


```

my $eps; # función dieléctrica del cilindro.
my $R=1; # Usa el radio como unidad de distancia.
my $c=1; # R/c como unidad de tiempo.
my ($wmin, $wmax, $Nw); # frecuencias mínima, máxima y número
my $lmax; # maximo momento angular
my $options=q(
'eps=f'=>\$eps, # función dieléctrica del cilindro
  'wmin=f'=>\$wmin, # frecuencia mínima
  'wmax=f'=>\$wmax, # frecuencia máxima
  'Nw=i'=>\$Nw, # número de frecuencias
  'lmax=i'=>\$lmax, # máximo momento angular
);
my %options=(eval $options);
die "Error en definición de opciones; $@" if $@;
GetOptions(%options) or usage($options, "Opciones erróneas");
usage($options, "Faltan parámetros")
  unless List::Util::all {defined $_} ($eps, $wmin, $wmax, $Nw, $lmax);
my $w=zeros($Nw)->xlinvals($wmin,$wmax); # Arreglo de frecuencias
my $qR=$w/$c*$R;
my $kR=sqrt($eps)*$w*$R;
my $l=sequence($lmax+1);

```

Figura 3: inicializa

Necesitamos una rutina para explicar el uso si hubiera un error.

```

sub usage {
  my ($options, $message)=@_;
  say $message if defined $message;
  say $options;
  exit 1;
}

```

Figura 4: uso

Calcula

```

# Calcula las funciones de Bessel, Neumann y Hankel
my $JlqR=bessjn($qR->dummy(1), $l->dummy(0));
my $NlqR=bessyn($qR->dummy(1), $l->dummy(0));
my $HlqR=$JlqR+i()*$NlqR;
my $JlkR=bessjn($kR->dummy(1), $l->dummy(0));

```

```

# Calcula sus derivadas. Itera sobre frecuencias. Pon l como 1er índice.
my $DJlqR=deriva($JlqR->mv(1,0), $qR->dummy(0))->mv(0,1);
my $DHlqR=deriva($HlqR->mv(1,0), $qR->dummy(0))->mv(0,1);
my $DJlkR=deriva($JlkR->mv(1,0), $kR->dummy(0))->mv(0,1);
# factor 1,2,2,2... para contar l's negativas
my $factor_l=(1+($l>=1))->dummy(0);
# Polarización TM
# Coeficientes
my $al_TM=-($eps*$qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)
           /($eps*$qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);
my $bl_TM=($JlqR+$al_TM*$HlqR)/$JlkR;
# Secciones
my $sigma_s_TM=($factor_l*4/$qR*$al_TM->abs2)->mv(1,0)->sumover;
my $sigma_ex_TM=($factor_l*4/$qR*$al_TM->re)->mv(1,0)->sumover;
# Campo
my $E2_TM=($factor_l*($kR/$eps*$bl_TM*$DJlkR)->abs2)->mv(1,0)->sumover;
# Polarización TE
# Coeficientes
my $al_TE=-($qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)
           /($qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);
my $bl_TE=($JlqR+$al_TE*$HlqR)/$JlkR;
# Secciones
my $sigma_s_TE=($factor_l*4/$qR*$al_TE->abs2)->mv(1,0)->sumover;
my $sigma_ex_TE=($factor_l*4/$qR*$al_TE->re)->mv(1,0)->sumover;
# Campo
my $E2_TE=($factor_l*($bl_TE*$JlkR)->abs2)->mv(1,0)->sumover;

```

Figura 5: calcula

Necesito una subrutina para derivar las funciones de Bessel. Uso la expresión recursiva que relaciona la derivada de un orden con el orden anterior. El caso $l=0$ es especial. Asumo que l es el primer índice. De haber más, itera automáticamente.

```

sub deriva { #l es el primer índice
  my ($B1, $x) = @_; # B(x,l), B is any Bessel, $x are arguments of Bessel
  my $B1m1=$B1->rotate(1); # B(x,l-1)
  my $DB1=-$B1->xvals/$x*$B1+$B1m1;
  $DB1((0)).=-$B1((1));
  return $DB1;
}

```

Figura 6: deriva

Por completez, va abajo la versión en términos de la siguiente en lugar de la previa función de Bessel. En este caso, la derivada del orden máximo está equivocada y no debe usarse.

```
sub deriva {
  my ($Bl, $x) = @_; # B(x,l), B is any Bessel, $x are arguments of Bessel
  my $Blp1=$Bl->rotate(-1); # B(x,l+1)
  my $DBl=$Bl->xvals/$x*$Bl-$Blp1; # don't trust highest l;
  return $DBl;
}
```

Figura 7: deriva-alt

Grafica

```
my $gw=gpwin();
$gw->plot({title=>'F. Bessel del primer tipo', xlabel=>'qR', yzeroaxis=>1},
  with=>'lines', $qR, $JlqR);
prompt -single, -void, "Listo?";
$gw->plot({title=>'Derivadas de Bessel del primer tipo', xlabel=>'qR', yzeroaxis=>1},
  with=>'lines', $qR, $DJlqR);
prompt -single, -void, "Listo?";
$gw->plot({title=>'F. Bessel del segundo tipo', xlabel=>'qR', yzeroaxis=>1, yrange=>[-
  with=>'lines', $qR, $NlqR, with=>'lines', $qR, pdl(0));
prompt -single, -void, "Listo?";
$gw->plot({title=>"Coeficientes de esparcimiento TM {/Symbol e}=$eps", xlabel=>'qR', y
  with=>'lines', $qR, $al_TM->abs2); # remueve la l más alta
prompt -single, -void, "Listo?";
$gw->plot({title=>"Secciones transversales TM {/Symbol e}=$eps", xlabel=>'qR', yzeroax
  with=>'lines', $qR, $sigma_s_TM,
  with=>'lines', $qR, $sigma_ex_TM,
);
prompt -single, -void, "Listo?";
$gw->plot({title=>"Campo cuadrado promedio TM {/Symbol e}=$eps", xlabel=>'qR', yzeroax
  with=>'lines', $qR, $E2_TM);
prompt -single, -void, "Listo?";
$gw->plot({title=>"Coeficientes de esparcimiento TE {/Symbol e}=$eps", xlabel=>'qR', y
  with=>'lines', $qR, $al_TE->abs2); # remueve la l más alta
prompt -single, -void, "Listo?";
$gw->plot({title=>"Secciones transversales TE {/Symbol e}=$eps", xlabel=>'qR', yzeroax
```

```

        with=>'lines', $qR, $sigma_s_TE,
    with=>'lines', $qR, $sigma_ex_TE,
);
prompt -single, -void, "Listo?";
$gw->plot({title=>"Campo cuadrado promedio TE {/Symbol e}=$eps", xlabel=>'qR', yzeroax
        with=>'lines', $qR, $E2_TE);
prompt -single, -void, "Listo?";

```

Figura 8: grafica

Finalmente, juntamos todo en un programa que grafica los espectros de los coeficientes de dispersión, las secciones transversales y el campo superficial.

```

#Calcula los coeficientes de esparcimiento de un cilindro dieléctrico.
<<paquetes>>
<<inicializa>>
<<calcula>>
<<deriva>>
<<grafica>>
<<uso>>

```

Figura 9: todo

6.2. Resultados

Corrimos el programa 9 como en la figura 10 indicando una función dieléctrica $\epsilon = 10$, y haciendo cálculos para 1000 frecuencias entre .1 y 3.5, empleando momentos angulares $l = 0, 1, \dots 10$.

```
./mie2D.pl -eps 10 -wmin .1 -wmax 3.5 -Nw 1000 -lmax 10
```

Figura 10: Ejecución del programa 9.

La figura 11 muestra las primeras funciones de Bessel del primer tipo. Son regulares en el origen, donde son aproximadamente potencias de su argumento, i.e., J_0 es constante, J_1 lineal, J_2 cuadrática y así sucesivamente. En la figura 12 mostramos las derivadas de las funciones de Bessel, para verificar cualitativamente que correspondan a lo esperado a partir de la fig. 11. En la figura 13 mostramos las primeras funciones de Bessel del segundo tipo de orden entero, o funciones de Neumann N_l . Observamos su singularidad




Figura 11: Primeras funciones de Bessel del primer tipo J_l con órdenes enteros $l = 0, 1, 2, \dots$




Figura 12: Derivadas J'_l respecto a su argumento de las primeras funciones de Bessel del primer tipo.




Figura 13: Funciones de Neumann N_l con órdenes enteros $l = 0, 1, 2 \dots$

en el origen. Con estas podemos obtener las funciones de Hankel salientes empleadas en el cálculo de esparsamiento.

Las figuras 14 y 15 muestra los coeficientes de dispersión correspondientes a las polarizaciones TM y TE, respectivamente correspondientes a un cilindro dieléctrico con permitividad $\epsilon = 10$ iluminado con una onda plana monocromática de frecuencia $\omega = qc$ de amplitud 1. Notamos que el valor máximo para los coeficientes de dispersión es 1. Esto se debe a que distintos momentos angulares no se mezclan debido a la simetría rotacional, y para cada valor l del momento angular, no puede esparcirse más energía que la que incide. Las formas de las curvas son curiosas y algunas muestran resonancias muy delgadas mientras que otras no.

Las figuras 16 y 17 muestran el cuadrado del campo eléctrico paralelo a la superficie promediado sobre ella para ambas polarizaciones. Notamos que aparecen una serie de picos angostos y altos. Estos corresponden a las resonancias de Mie 2D. Notamos que para toda l , para distancias suficientemente largas el campo esparcido se vuelve propagante. Por tanto, siempre hay pérdida de energía de estos modos, aunque los materiales mismos sean dieléctricos que no presenten absorción. Por tanto, las resonancias de Mie




Figura 14: Coeficientes de dispersión a_l evaluados en el exterior de un cilindro dieléctrico con permitividad $\epsilon = 10$ como función de la frecuencia normalizada qR .




Figura 15: Coeficientes de dispersión como en la fig. 14 pero para polarización TE.




Figura 16: Cuadrado del campo electrico tangencial E_θ^2 promediado sobre la superficie del cilindro para polarización TM.




Figura 17: Cuadrado del campo electrico tangencial E_z^2 promediado sobre la superficie del cilindro para polarización TE.

tienen necesariamente anchos y alturas finitas. Los anchos disminuyen conforme aumenta el índice de la resonancia, hasta llegar a la resonancia situada alrededor de $qR \approx 2,7$. Para polarización TM el campo alcanza amplificaciones mayores que para polarización TE.

Finalmente, en las figuras 18 y 19 mostramos las secciones de esparcimiento y de extinción, σ_{es} y σ_{ex} para ambas polarizaciones. Notamos que no hay diferencia apreciable entre ellas, pues no introdujimos disipación. La forma de las curvas se ve bastante caprichosa.




Figura 18: Secciones transversales σ_{es} y σ_{ex} para polarización TM.

6.3. Campo

Ahora presento un código para la visualización de los campos. Para cargar los paquetes y las opciones usuales reciclo el código de la fig. 2. Los parámetros son ahora diferentes.

```
my $eps; # función dieléctrica del cilindro.
my $R=1; # Usa el radio como unidad de distancia.
my $c=1; # R/c como unidad de tiempo.
my $w; # frecuencias para el cálculo
```




Figura 19: Secciones transversales σ_{es} y σ_{ex} para polarización TM.

```

my $lmax; # maximo momento angular
my $xmax; # x o y máximas para graficar, en unidades del radio R
my $N; # 2N+1 es el número de pixeles en x y en y
my $Rcut=$R; # cut plot before this radius (default 1)
my $options=q(
'eps=f'=>\$eps, # función dieléctrica del cilindro
'w=f'=>\$w, # frecuencia
'lmax=i'=>\$lmax, # máximo momento angular
'xmax=f'=>\$xmax, # x o y distancia máximas para graficar, en unidades del radio R
'N=i'=>\$N, # 2N+1 es el número de pixeles en x y en y
'Rcut=f'=>\$Rcut; # cut plot before this radius
);
my %options=(eval $options);
die "Error en definición de opciones; $@" if $@;
GetOptions(%options) or usage($options, "Opciones erróneas");
usage($options, "Faltan parámetros")
unless List::Util::all {defined $_} ($eps, $w, $lmax, $xmax, $N);
my $qR=$w/$c*$R;

```

```

my $kR=sqrt($eps)*$w*$R;
my $l=sequence($lmax+1);
my $cartesian=(zeroes(2*$N+1,2*$N+1)->ndcoords-$N)/$N*$xmax; # cartesian coordinates
my $r=$cartesian->abs2->sumover->sqr; # magnitudes
my $theta=atan2($cartesian((1)), $cartesian((0))); # angles

```

Figura 20: inicializa1

En caso de errores, reciclamos la rutina de la fig. 4. Calculamos los coeficientes de dispersión de una manera similar a la fig. 5, pero para una sola frecuencia.

```

# Calcula las funciones de Bessel, Neumann y Hankel en la superficie
my $JlqR=bessjn($qR, $l); # l
my $NlqR=bessyn($qR, $l);
my $HlqR=$JlqR+i()*$NlqR;
my $JlkR=bessjn($kR, $l);
# Calcula sus derivadas.
my $DJlqR=deriva($JlqR, $qR);
my $DHlqR=deriva($HlqR, $qR);
my $DJlkR=deriva($JlkR, $kR);
# factor 1,2,2,2... para contar l's negativas
my $factor_l=(1+($l>=1));
# Polarización TM
my $al_TM=-($eps*$qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)
           /($eps*$qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);
my $bl_TM=($JlqR+$al_TM*$HlqR)/$JlkR;
# Polarización TE
my $al_TE=-($qR*$DJlqR*$JlkR-$kR*$JlqR*$DJlkR)
           /($qR*$DHlqR*$JlkR-$kR*$HlqR*$DJlkR);
my $bl_TE=($JlqR+$al_TE*$HlqR)/$JlkR;

```

Figura 21: coeficientes

Puedo reciclar el código de la figura 6 para calcular las derivadas de las funciones de Bessel. Ahora puedo calcular el campo esparcido y su cuadrado, proporcional al flujo de energía en la zona de radiación. Lo escalo con r para que a largas distancias coincida con la sección diferencial de dispersión.

```

my $il=($l%4==0)+i()*($l%4==1)-($l%4==2)-i()*($l%4==3); # [1] i^l
my $m1l=2*($l%2==0)-1; # [1] -1^l
my $eiltheta=exp(i()*$theta->dummy(0)*$l); # [1,nx,ny]

```

```

my $Jlqr=bessjn($qR*$r->dummy(0), $l); # [l,nx,ny]
my $Nlqr=bessyn($qR*$r->dummy(0), $l); # [l,nx,ny]
my $Hlqr=$Jlqr+i()*$Nlqr; # [l,nx,ny]
# use  $H_{-1}=H_1$ ,  $i^{-1}=(i^1)^*$  and  $e^{-il\theta}=(e^{il\theta})^*$ 
#  $a_{-1}=a_1$ 
# TM
my $H_TM=($il*$al_TM*$Hlqr*$eiltheta->sumover +
           ($il->conj*$al_TM*$m1l*$Hlqr*$eiltheta->conj)->(1:-1)->sumover;
my $S_TM=$H_TM->abs2*($r>$Rcut);
# TE
my $E_TE=($il*$al_TE*$Hlqr*$eiltheta->sumover +
           ($il->conj*$al_TE*$m1l*$Hlqr*$eiltheta->conj)->(1:-1)->sumover;
my $S_TE=$E_TE->abs2*($r>$Rcut);

```

Figura 22: poynting

Grafica los resultados.

```

my $gw=gpwin();
$gw->plot({title=>"r*H2/H2i TM {/Symbol e}=$eps {/Symbol w}=$w ",
           xlabel=>'x', ylabel=>'y', justify=>1},
          with=>'image', $cartesian((0)), $cartesian((1)), $r*$S_TM);
prompt -single, -void, "Listo?";
$gw->plot({title=>"r*E2/E2i TE {/Symbol e}=$eps {/Symbol w}=$w ",
           xlabel=>'x', ylabel=>'x', justify=>1},
          with=>'image', $cartesian((0)), $cartesian((1)), $r*$S_TE);
prompt -single, -void, "Listo?";

```

Figura 23: grafical

Finalmente, juntamos todo en un programa que grafica la intensidad esparcida.

```

#Calcula los coeficientes de esparcimiento de un cilindro dieléctrico.
<<paquetes>>
<<inicializa1>>
<<coeficientes>>
<<poynting>>
<<grafical>>
<<deriva>>
<<uso>>

```

Figura 24: todo1

6.4. Resultados

Corrimos el programa 24 como en la figura 25 indicando una función dieléctrica $\epsilon = 10$, y haciendo cálculos para diversas frecuencias $\omega R/c = 0,1, 2,319, 2,31, 1,57$, con una $l_{\max} = 10$.

```
./poynting2D.pl -eps 10 -w 0.1 -lmax 10 -xmax 5 -N 100
./poynting2D.pl -eps 10 -w 2.319 -lmax 10 -xmax 5 -N 100
./poynting2D.pl -eps 10 -w 2.31 -lmax 10 -xmax 5 -N 100
./poynting2D.pl -eps 10 -w 1.57 -lmax 10 -xmax 5 -N 100
```

Figura 25: Ejecución del programa 24.

En la gráfica 26 mostramos el campo magnético al cuadrado normalizado rH^2/H_i^2 con la distancia r y el campo H_i de la onda incidente, proporcional a distancias grandes a la sección diferencial de esparcimiento, para una frecuencia pequeña $\omega = 0,1c/R$ y para polarización TM.




Figura 26: Cuadrado del campo H normalizado al campo incidente H_i y escalado con r para polarización TM y para una frecuencia baja $\omega = 0,1$.

Se ve claramente un patrón de campo cercano de caracter dipolar y a lo lejos un patrón de radiación radiación dipolar con el dipolo orientado a lo

largo de la dirección vertical, que es la dirección del campo incidente que se propaga hacia la derecha. Como estoy multiplicando el cuadrado del campo por la distancia r , el valor asintótico sólo depende de la orientación.

En la gráfica 27 la curva correspondiente para polarización TE. En este




Figura 27: Cuadrado del campo eléctrico, como la figura 26 pero para polarización TE.

caso el campo eléctrico apunta en la dirección del eje, induciendo un dipolo axial que radía isotrópicamente alrededor. Como para esta polarización y en el límite cuasiestático no hay campo de depolarización, el dipolo inducido y la radiación son mayores.

La figura 28 corresponde a una frecuencia relativamente alta $\omega = 2,319c/R$ que es además una de las resonancias altas y angostas de las figuras 14 y 16. Notamos claramente un patrón de radiación con 8 lóbulos, correspondiente a una resonancia octopolar $l = 4$. Si disminuimos ligeramente la frecuencia, por ejemplo, a $\omega = 2,31c/R$, (fig. 29 la intensidad disminuye notablemente y se vuelve más visible la dispersión frontal, responsable de extinguir al haz incidente, de acuerdo al teorema óptico.

Algo similar sucede cerca de las resonancias para polarización TE, como muestra la figura 30 calculada a $\omega = 1,57c/R$ para polarización TE. Se ob-




Figura 28: rH^2/H_i^2 , como en la figura 26 para en una frecuencia $\omega = 2,319c/R$ cercana a una resonancia.




Figura 29: rH^2 , como en la figura 28 para en una frecuencia ligeramente menor $\omega = 2,31c/R$.




Figura 30: rE^2 , como en la figura 28 pero para polarización TE a una frecuencia $\omega = 1,57c/R$ cercana a una resonancia.

serva claramente un patrón de 6 lóbulos correspondiente a una resonancia hexapolar con $l = 3$.

7. Conclusiones

Desarrollamos la teoría para el cálculo del esparcimiento por cilindros dieléctricos en términos de funciones de Bessel. Obtuvimos la amplificación del campo en la superficie, la cual puede ser un par de órdenes de magnitud mayor que el campo incidente, y mostramos las resonancias tipo Mie y su manifestación en las secciones transversales de dispersión.

8. Apéndice. Gaussiana imaginaria

En este apéndice mostramos como obtener la integral (31). Para ello, consideremos primero la integral

$$I = \int_C d\zeta e^{i\zeta^2}, \quad (44)$$

sobre el circuito cerrado mostrado en la fig. 31. Como no hay singularidades en su interior, la integral es $I = 0$. Por otro lado, la integral puede separarse en tres partes como

$$\begin{aligned} 0 = I &= \int_0^R d\zeta e^{i\zeta^2} + \int_0^{\pi/4} d(Re^{i\theta}) e^{i(Re^{i\theta})^2} - \int_0^R d(t\sqrt{i}) e^{i(t\sqrt{i})^2} \\ &= \int_0^R d\zeta e^{i\zeta^2} + \int_0^{\pi/4} d\theta R e^{i\theta} e^{iR^2 e^{i2\theta}} - \sqrt{i} \int_0^R dt e^{-t^2}. \end{aligned} \quad (45)$$

El segundo término incluye un factor de la forma $R \exp(iR^2(\cos 2\theta + i \sin 2\theta)) = R \exp(iR^2 \cos 2\theta) \exp(-R^2 \sin 2\theta)$. Para $\theta \in (0, \pi/4)$, $\sin 2\theta > 0$ y ese factor tiene magnitud menor a 1 y tiende a cero rápidamente cuando $R \rightarrow \infty$. Por lo tanto, en el límite $R \rightarrow \infty$ podemos escribir

$$\int_0^\infty d\zeta e^{i\zeta^2} = \sqrt{i} \int_0^\infty dt e^{-t^2} = \frac{\sqrt{i}\pi}{2}, \quad (46)$$

de donde el resultado se sigue de inmediato.

9. Apéndice. Códigos de metapost

Código para producir las figuras 1 y 31.




Figura 31: Circuito \mathcal{C} para realizar la integral (44) consistente en una recta de 0 hasta R sobre el eje real, un octavo de círculo de radio R y el retorno recto al origen.

```

g_use_svg = 0; %0 for postscript, 1 for svg
prologues:=3;
u=1cm;

outputtemplate := "%j-%c.eps";
if g_use_svg > 0:
    outputtemplate := "%j-%c.svg";
    outputformat := "svg";
fi

color yellow;
yellow=green+red;

beginfig(1)
    path p[];
    p0=fullcircle xscaled 3u yscaled 1u;
    p1=p0 shifted (10u*up);
    p2=subpath (-4,0) of p0--subpath(0,-4) of p1--cycle;
    fill p2 withcolor .5blue+.5white;
    fill p1 withcolor .7blue+.3white;
    p3=unitsquare xscaled 10u yscaled 2u slanted -.5 rotated 90;
    for i=-15 upto -10:
        fill p3 shifted (i*((.5u,0))rotated -10) withcolor yellow;
        draw p3 shifted (i*((.5u,0))rotated -10);
    endfor;
    drawarrow (((0,0)--(2u,0)) shifted (-10*((.5u,0))) rotated -10
        shifted .5[point 0 of p3, point 2 of p3] withpen pencircle scaled
        2pt;
    label.top(btex $\vec q$ etex scaled 2, (1.5u,0) shifted (-10*((.5u,0))) rotated -10 s
        of p3, point 2 of p3] withpen
        pencircle scaled 2pt;
    drawarrow (((0,0)--(2u,0)) shifted (10*((.5u,0))) rotated -10 shifted
        .5[point 0 of p3, point 2 of p3] withpen pencircle scaled 2pt;
    drawarrow (((0,0)--(2u,0)) shifted (10*((.5u,0))) rotated -30 shifted
        .5[point 0 of p3, point 2 of p3] withpen pencircle scaled 2pt;
    drawarrow (((0,0)--(2u,0)) shifted (10*((.5u,0))) rotated 10 shifted
        .5[point 0 of p3, point 2 of p3] withpen pencircle scaled 2pt;
    drawarrow ((0,0)--(0,1u)) shifted(0,-2u);
    drawarrow ((0,0)--(1u,0)) rotated -10 shifted (0,-2u);
    drawarrow ((0,0)--(0,-.75u)) slanted -.5 rotated 90 shifted (0,-2u);

```

```

label.ulft(btex  $z$  etex scaled 2, (0,1u) shifted (0,-2u));
label.lrt(btex  $x$  etex scaled 2,(1u,0) rotated -10 shifted (0,-2u));
label.urt(btex  $y$  etex scaled 2, (0,-.75u) slanted -.5 rotated 90 shifted (0,-2u));
interim ahangle:=180;
drawdblarrow ((0,0)--(point 0 of p0))shifted (0,11u);
label.top(btex  $R$  etex scaled 2, (point 0 of p0) scaled .5 shifted (0,11u));
endfig;
beginfig(2)
  path p[];
  p0=(0,0)--(5u,0)..subpath(0,1) of fullcircle scaled 10u--cycle;
  drawdblarrow (0,7u)--(0,0)--(7u,0);
  label.lrt(btex  $\operatorname{Re}\zeta$  etex scaled 1.5, (7u,0));
  label.ulft(btex  $\operatorname{Im}\zeta$  etex scaled 1.5, (0,7u));
  label.bot(btex  $0$  etex scaled 1.5, (0,0));
  label.bot(btex  $R$  etex scaled 1.5, point 1 of p0);
  drawarrow subpath(0,.5)of p0 withpen pencircle scaled 2pt;
  drawarrow subpath(.5,2.5)of p0 withpen pencircle scaled 2pt;
  drawarrow subpath(2.5,3.5)of p0 withpen pencircle scaled 2pt;
  draw subpath(3.5,4)of p0 withpen pencircle scaled 2pt;
endfig;
end.

```

Este código se puede correr con el comando

`mpost notas`